
Salvage Procedures using GHS - 4/2011

Scenarios encountered in salvage operations are a challenge with many factors affecting how to 
proceed.  Fire, rescue of personal, weather conditions, location, equipment available are some of 
these factors.  The job of the naval architect during a rescue is to provide information needed by 
the operators.  Ground reactions, forces needed to lift or roll, amount of floatation devices, effects 
of deballasting, offloading or shifting cargo and longitudinal strength are some of the areas of 
concern.  Of course, no book or manual will be able to cover every possible scenario.  This 
manual is to demonstrate some of the features and capabilities of GHS that may be useful to a 
naval architect involved with salvage.

It is assumed the user has a good understand of and experience using GHS.  Basic concepts, 
understanding of the geometry model, commands for straight forward stability calculations and 
the command syntax and working in the GHS environment  should be familiar to the user.  In 
addition, macros, passing parameters and user variables will be used throughout this manual and 
the user should have some experience with them.  The additional modules needed, and provided 
in the Salvage bundle, are Condition Graphics, Load Editor and Longitudinal Strength.

Geometry Model
If you are lucky, a geometry file exists and is available.  If not, one will need to be created.  If you 
have created geometry files in the past, you are probably aware that there are many ways to do 
so.  If there are other programs that you are more familiar with that can create a .GF file directly 
or export to a format Model Converter accepts, that may be the approach to take.  Keeping in 
mind that many factors are involved in salvage work, such as estimations, approximations and 
variables beyond human control.  As time is usually critical, the geometry file does not need to be 
stability analysis ready.  The tanks to be used for ballast or are flooded can be modeled as 
needed.  If the tank capacity is known, a simple block shape in the general location may be 
suitable.  Accounting for tank loads as point loads and applying free surface corrections will save 
time.  Sail parts can be simple rectangles.  Superstructure modeled as a sail part may need to be 
remodeled as a displacer if submerged and providing buoyancy.

Geometry modeling will not be discussed in this manual other that stated above.  For the 
scenarios presented below, the standard models, FV.GF, SV.GF and CSI100.GF will be used.

Grounding with severe heel
This section demonstrates the recovery of a grounded hard chine vessel that is on its side.  How 
to setup the initial condition, define ground points, determine the force needed to right the vessel 
and finding the maximum ground reactions will be discussed.  Fig. 3 shows the initial condition.

WEIGHT
One piece of information needed is the weight or displacement and center of gravity at the time of 
the grounding.  If draft and trim is known prior to grounding, that information can be used to get 
the displacement and LCG.  VCG may be obtained from a stability book or ship's log if available. 
For this exercise, a lightship weight, center and loads are assumed to be known.  When taking 
information from the ship's log or pre casualty condition, a correction for any shifted cargo will be 
necessary.

WATERPLANE
To establish the grounded condition, the waterplane and ground points need to be defined.  Let's 
consider setting the waterplane first.  In cases such as this, draft mark readings are unavailable. 
The DRAFT command is not practical.  There are two approaches to defining the waterplane.  In 
GHS, the trim expressed in degrees is the arc tangent of the distance between the waterline at 
two longitudinal locations divided by the distance between the two locations.  

TRIM = ATAN (1.66/(45 – 20)) = 3.80°

        Figure 1

This can be determine from observations in the field.  In this example, notice that the keel is 
exposed and it is parallel to the baseline.  By measuring the height of the keel above the water at 
two locations and the distance between them, the trim angle can be calculated as illustrated 
above.  The heel can be approximated by visual inspection.  Lastly, the depth is set by giving the 
height of a known point on the vessel.  To do this, a critical point must be define at the point in 
question.  We will use the heel of the skeg and set the height to that observed.

CRTPT (4) "end of skeg" 30.8a 0 0 /noflood
HEIGHT (4) 2
An easier method of defining a ground plane, even for extreme attitudes, is to use the WPL3 

wizard.  The wizard will prompt the user for the location of three points on the vessel and the 
height of the points above the waterplane.  Then designate the three points as either port,stbd, 
fwd or aft.  Then the wizard will solve for the correct depth, trim and heel for the heights given and 
report any difference.  If the differences are large, the most common cause is improper assigning 
of points to locations.  The convergence factors can be adjusted for small changes.  The range is 
0.01 to 2.0 for both trim and heel convergence factors.

Once the waterplane is set, ground points can be defined.  In fact,  the depth is required before 
any ground points can be defined.  This is needed so the waterplane and the point of contact, or 
"the bottom", have a reference to one another via the ground point on the vessel and the given 
penetration.  Doing so allows simulation of rising or falling tides.

GROUND POINTS
Defining the number and location of ground points and adjusting the input parameters will require 
some judgment.  On sand or mud bottoms, an array of ground points may be needed to 
reasonably determine loads on the hull.  Rocks or reefs would necessitate fewer ground points. 
An underwater survey will assist in determining the location of ground points and the condition of 
the bottom.

For this example, the ground points supporting the vessel are:

GROUND "Fwd chine" * 27.0f 8.42 4.36 
GROUND "Fwd hull side" * 27.0f 10.00 10.00
GROUND "Aft chine" * 23.1a 11.63 4.78
GROUND "Aft hull side" 23.1a 11.93 8.89

As the vessel is righted, the hull may come in contact with the ground at different locations.  

Figure 2

Therefore, additional ground points are needed.  Defining these ground points can be 
complicated due to variations in the bottom topography, hull shape and the uncertainty of the 
attitude the vessel will take during the rolling procedure.  Two contact points are assumed in this 
case.  One at the bow and one on the skeg.  Since they are not in contact at this stage, the 
penetration is set at a negative value equal to the vertical distance from the ground point to the 
bottom where contact is expected.

GROUND "Keel at bow" * 23.0f 0 0 /pen:-11.5
GROUND "End of skeg" * 30.8a 0 0 /pen:-10.0

Now equilibrium and the reactions can be found by solving and viewing the status.

SOLVE
STATUS GHS
STATUS CRT

If a Condition Graphic view is desired, the following command will produce the image in Figure 3.

DISP (*) STATUS BODY @27f @23.1a @30.8a, PROFILE: OUTBOARD

Figure 3

In the body view at 30.8a (top right), the anticipated ground point at the end of the skeg is 
displayed as a small circle and a vertical line extends from that to a green triangle representing 
the depth to the bottom.  As the vessel rolls, the green triangle will be displayed in a different 
horizontal position but will remain directly under the end of the skeg.

PULLING
The PULL command can be used to model the effect of cables attached to the vessel and tension 
applied.  Within GHS, the magnitude of the force, the located of attachment and its direction is 
converted into heeling and trimming moments.  The elevation angle is relative to the waterplane. 
The azimuth angle is measured from the transverse axis.  An angle of 0 is to starboard when the 
vessel is upright and will result in a heeling moment to starboard.  If the vessel heels past 90 
degrees, the pull will then induce a heeling moment to port.  Consider the vessel oriented in a 
north-south direction with the bow heading north.  An azimuth angle of 0 degrees implies a pull 
force to the east and remains in that direction regardless of the heel of the vessel.  The arm is the 
distance from the point of attachment to the centroid of the submerged lateral plane taking into 
account the elevation and azimuth angle.  

Since the amount of force needed is unknown, the pull can be initially set as zero.

PULL "WINCH LINE AT BOW" 0 30.8f 12.0p 21.2 180 -5
PULL "WINCH LINE AT STERN" 0 23.1a 12.25 13.55 180 -5

The above defines the attach point at the deck edge.  The azimuth value of 180 sets the direction 
of the force to port.  The elevation of -5 is relative to the waterplane and is downward in this case. 
Given the length of leads in salvage operations, this can be considered constant with little effect 
on the results.

LOAD EDITOR for Windows (LEw)
Entering the load editor provides the user the ability to analyze the problem interactively.  There 
are many parameters available when calling Load Editor.  For our purposes, we need to define 
some macros and variables to be used by Load Editor.

MACRO REHEEL "RESETHEEL"
  HEEL 75
  SOLVE
/

MACRO HMMTPULL "PULL SUMM"
  HMMT REPORT
  WAIT
  PULL
  WAIT
/

Since we want to use Condition Graphics, we have to define the views to be displayed by setting 
the variable cgparam as

VARIABLE CGPARAM = "BODY @27F @23.1A, BODY @23.0F @30.8A, 
PROFILE:OUTBOARD /SYNC"

To finally launch Load Editor,

LOAD (*) EDIT /DISPLAY:OFF  /CG  /PULL  /MACRO:REHEEL, HMMTPULL

The pull window can be used to applied a pulling force.  If a value is applied that "tips" the vessel, 
the ResetHeel button can be used to return to the starting condition and smaller forces applied. 
GHS does not consider horizontal forces.  The effect of the pull command is treated as heeling 
moment.  To see the resulting heeling moment and pull forces that produces that moment, click 
the Winch button.

Starting Load Editor will bring up a window similar to Figure 4.

Figure 4

The "Tank", "Weights", "Ground" and "Pull" buttons will bring up their respected windows.  Tank 
loads, weights and the pull forces can be changed and the results displayed after clicking the 
"Solve" button.  The "Auto" button can be clicked to eliminate the need to click "Solve".

Once enough pull force is applied to right the vessel, the Condition Graphic window displays the 

vessel in nearly an upright position.  Note that there is some resulting reactions at the skeg.  What 
may not be obvious is the pull forces are still applied.  If the lines go slack during the rollover, 
GHS does not take that into account.  It is important to understand that once the pull force is 
define, it is in effect until the user deletes it.

GROUND REACTIONS
Up to this point we only know what will happen up to the tipping point.  During the continuation of 
the rollover, the lost buoyancy due to the emerging topsides or superstructure may be greater 
than the gained buoyancy due to the submerging hull bottom.  This could effect the ground 
reactions in unpredictable ways.  To find the maximum ground reaction, a short macro can step 
through the rolling at small increments and the ground reactions evaluated.

VARIABLE HE MAXHEEL MAXREACT = 0
.REHEEL
FIX HEEL

MACRO TIPOVER
  SET HE = {HEEL} MIN 1
  HEEL {HE}
  SOLVE
  IF {REACT} = 0 THEN EXIT
  IF {REACT} > {MAXREACT} THEN SET MAXREACT = {REACT} | SET MAXHEEL = 
{HEEL}
  \TOTAL GROUND REACTION:{ REACT:2},  RA ={ RAH:2},  GMT = { GMT:2}, HEEL ={ HEEL}
/

To run the macro and see the reactions at every ground point at the maximum

.TIPOVER (90)
\\\
HEEL {MAXHEEL}
SOLVE
\THE MAXIMUM GROUND REACTIONS OCCURRED AT {HEEL} DEGREES\
\\
STATUS DISPL

The above approach is a check of the total ground reaction.  The reactions at the two chine 
locations are reported when the total reaction is at its maximum.  It is possible that these 
individual reactions may be greater at some other heel angle that at the angle of the maximum 
total ground reaction.

STABILITY THROUGH THE ROLLOVER
To see the righting arm during the roll, return to the initial condition and change the angles to 
produce the curve as the vessel rolls to port.

.REHEEL
VARY heel
SOLVE
ANGLES 0 -5 -10 ... -90

The righting arm curve should look similar to Figure 5.

Figure 5

A copy of the run file to perform the above analysis (and following examples) can be found in the 
Appendix.  The commands can be copied and pasted into a text file.

Capsized Vessel
The pull command can be used to find the forces needed to right a capsized vessel.  Since the 
command converts the defined pull force into a moment, the total moment needed to roll the 
vessel can be determined by a single pull command.  In reality, bridles and cables would be 
arranged to pull in opposite directions at two or more locations.  In the example that follows, only 
one location will be used, 45 feet fwd of the origin. 

From the geometry model SV.GF the following hull offsets are taken at station 44.87f.

Figure 6
In the previous example, the pull command applied a given force at the attach point for all angles 
of heel.  In this case, the force applied will be a function of the heel angle.  This will simulate the 
"unfurling" of the bridle as the vessel rolls.  Assuming the pull force is directed 10 degrees 
downwards relative to the waterplane, and starting at a heel of 180 degrees, the force will be 
applied at the first point from 180 to 170 degrees.  Then the force is applied at the second point 
from 170 to 150 degrees.  And so on.  Using the above geometry, the commands would look like 
this

PULL "FWD DECK STBD" %1@180 %1@172 0@171  0@0   -45  6.39  9.36 180 10
PULL "FWD DECK CL" 0@180 0@171 %1@170 %1@168 0@167 0@0 -45  0     9.60 180   10
PULL "FWD DECK PORT" 0@180  0@168  %1@167 %1@0   -45 -6.39 9.36 180 10

PULL "FWD AT KEEL 1" %1@180   %1@170    0@169  0@0  -45  -1  0 0 -10
PULL "FWD BOTT 2" 0@180 0@170 %1@169 %1@150 0@149 0@0 -45  1  0 0 -10
PULL "FWD BOTT 3" 0@180 0@150 %1@149 %1@122 0@121 0@0 -45  2.28  0.47 0 -10
PULL "FWD BOTT 4" 0@180 0@122 %1@121 %1@108 0@107 0@0 -45  3.53  1.86 0 -10
PULL "FWD BOTT 5" 0@180 0@108 %1@107 %1@98  0@97 0@0  -45  4.25  3.15 0 -10
PULL "FWD AT DECK 6" 0@180    0@98      %1@97  %1@0  -45  6.39  9.36 0 -10

For the initial condition of a 2000 mt displacement, an LCG located at 27 m fwd of the origin and a 
VCG of 5 m, sample results for a pull of 250 mt and graphs of the righting arms and heeling arms 
for 50, 150, 250 and 290 mt forces are shown on the following pages.
The maximum force needed to overcome the maximum negative righting arm is approximately 
290 mt.  If a force is applied that is greater than the righting moment, the righting arm and heeling 
arm curves do not intersect.  In reality, a pull of that magnitude could never be achieved.  Once 
the vessel is rolled beyond the angle of largest negative righting arm, the pull forces will 
decrease.  However, the range of negative stability continues until 60 degrees, after which the 
vessel will continue to right itself.

Sunken Vessel

Using the pull command is not appropriate for raising vessels completely submerged.  However, it 
would be appropriate for rolling a submerged vessel that may be on its side.  For instance, to 
access a damaged area for repairs or to orient the vessel for lifting in an upright condition.

When a PULL is applied to the model, GHS treats the vertical component of the force as a point 
load.  A positive weight if the component is downwards, and a negative weight if the pull is 
upwards.  The horizontal component, if any, is handled as a moment.  The moment being equal to 
the magnitude of the horizontal component times the distance to the centroid of the submerged 
lateral plane area.

Consider a case where the sunken vessel is submerged and resting on the ground.  The total 
reaction of ground points will be the weight of the vessel (fix weights and tank loads) less any 
buoyancy.  When pull forces are applied, the vertical components will reduce the ground 
reactions.  When the total of pull forces exceeds the total of ground reactions, GHS will solve for 
the condition where any excess pull forces will be equal to the lost buoyancy due to the hull 
emerging from the water.  This may be acceptable if the only purpose is to determine the lift 
needed to raise the vessel.  If in this case, the bottom is an irregular surface and the vessel 
cannot be lifted straight up, some rolling will occur in the initial stages and if the ground reactions 
are of concern, the PULL command may not be suitable.  In addition, if the lift involves more than 
one lifting device, as the vessel rolls and gains and/or loses ground points, the force on the lifting 
devices will vary.   

Since such a procedure will be influenced by many factors, using the Multi Body module will 
provide more flexibility to the Naval Architect and provide more information and a better 
understand of the forces encountered and how they will interact.

GETTING STARTED WITH MULTI BODY
If you are familiar with Multi Body or have used it in the past, you may wish to skip to the next 
section.

Windows allows more than one session of GHS to run at the same time.  The Multi Body module 
allows these sessions of GHS to communicate with each other.  Multi Body ( from here on 
referred to as MB) operation is initiated by a special form of the solve command.  This is:

SOLVE PRIMARY ...parameters
or
SOLVE SECONDARY ...parameters
or
SOLVE NORMAL

Before discussing the parameters, understanding how the individual sessions of  GHS 
communicate with each other is needed.  Each of the sessions of GHS in use during MB 

operation, is considered either a primary or a secondary session.  Each primary may be 
connected to one or more secondaries, and each secondary may be connected to one or more 
primaries.  However, primaries may not be connected to primaries, and secondaries may not be 
connected to secondaries.  This is accomplished through "channel files".  Each primary session 
will have a unique channel.  Secondaries can have have one or more channels associated with 
them.  Once connected, the various sessions of GHS will communicate with each other though 
connection points defined through critical points.

There are two types of connection between these critical points, latched and grounded. 
Considering the case of an integrated tug-barge, latch type connections are needed.  Grounded 
type connections are used where one vessel is resting on another.  In this case, one vessel will 
exert an upward force only on the other, such as a drydock raising a vessel would.  If the 
connection points are not in contact, then no forces are exerted.  These critical points are defined 
in the coordinate system of the geometry model used in that session.  In addition, if two critical 
points are present, the 3 dimensional distance between them should be the same.  If three or 
more critical points are present, the distance between them and their relative locations to each 
other should be the same, since this type of connection requires that the critical points must be 
coincidental at all times.   The effect of the connection points not being coincidental will be 
discussed later.

Only vertical forces are considered in the interactions between the vessel, so it is best to think of 
the connections as ball and socket joints as opposed to pinned joints.

For the sake of simplicity, lets use a barge train consisting of two identical barges connected at 
the corners.  The geometry file CSI100.gf is a 100' x 40' x 10' rectangular barge with 4 equal port 
and starboard tank pairs.  As each session uses a single geometry file, however the same 
geometry file can be used.  Figure 7 shows the arrangement and connection details of the barge 
train.

Figure 7

The critical points for the forward barge would be:

CRTPT (1) "STBD PIN" 102.5 17.5 14.5 /NOFLOOD
CRTPT (2) "PORT PIN" 102.5 -17.5 14.5 /NOFLOOD

and for the aft barge:

CRTPT (1) "STBD PIN" -2.5 17.5 14.5 /NOFLOOD
CRTPT (2) "PORT PIN" -2.5 -17.5 14.5 /NOFLOOD

The transverse location of 17.5 feet off centerline is the location where the reaction forces are 
balanced.  The location of the actual reactions should not be used.  For example, 16.25 feet of 
centerline for the forward barge and 18.75 feet for the aft barge.  We will see why at the end of 
the exercise.

Once defined, setting the channel of communication and connecting point interactions would be 
done with the following commands.

SOLVE PRIMARY:TRAIN.MB /WAIT

and

SOLVE SECONDARY:TRAIN.MB,1,1; *,2,2 /WAIT

In the first statement above, TRAIN.MB  is the name given to the communication channel to be 
used.  If more than one primary session is to be used, a different channel name must be used in 
the statement for that session.  The parameter /WAIT tells this session of GHS to wait until 
communication is established with the secondary session. 

In the second statement, the critical points are connected by reference to their number, thus 1 to 
1 and 2 to 2.  The first in the pair refers to the critical point on the primary vessel, the second in 
the pair to the secondary vessel.  Since both critical point numbers are positive in each pair, this 
becomes a latched connection.  If one was negative, then it would be a grounded connection with 
that point being one that would not bear weight or not exert an upward reaction on the other 
vessel.  The * before the second pair indicates that the same communication channel is to be 
used.  If this secondary vessel was also connection to a different primary vessel, then the different 
communication channel would be stated.

So, in order to initiate the MB process, the following commands for the forward (primary) vessel 
would be:

CLEAR
READ CSI100.GF

DR 3
SO WE LCG
VCG 6

CRTPT (1) "STBD PIN" 100 20 10 /NOFLOOD
CRTPT (2) "PORT PIN" 100 -20 10 /NOFLOOD

SOLVE PRIMARY:TRAIN.MB /WAIT

ST GHS
ST CRT

LOAD (*) EDIT /DISPLAY:TRUE

the commands for the aft (secondary) vessel would be the same with the CRTPT and SOLVE 
commands as given on the previous page.  The LOAD (*) EDIT command enters a load editor 
window for each session with  MB running under load editor.  Any change in loads made in one 
load editor window  will be reflected in the other.

Exercise:
Write two run files using the above commands.  Start two sessions of GHS and execute each run 
file.  

What would happen if the connection point of one barge was 20 feet off centerline and for the 
other barge, 15 feet off centerline?  This is an exaggeration of using the location of the actual 
hinges on each barge.  The reactions computed by GHS are equal in magnitude but because the 
location, and thus the lever arms are different, the barges will result in different heels.

What would happen if the connection points were defined as being at the fwd end of each barge? 
Since we visualize the forward direction of model towards the right, we might visualize this 
arrangement as one barge on top of, or superimposed on the other.  GHS would find equilibrium 
but does not recognize any interference between the two hulls.  So if the Multi body sessions 
were set up in this way, loading Tank1 of either or both barges would result in both barges being 
trimmed by the bow.  

Note also, since GHS does not recognize any interference between the two hulls, a condition with 
excessive trim could result in the hulls making contact.  This is true regardless of the 
arrangement.  To prevent this, a combined trim (aft barge trim minus fwd barge trim) of 

Page 17 of 48



Salvage Procedures using GHS - 4/2011

2 x arctan (2.5/14.5) = 19.56 degrees

in this case would have to be an operational limit.

Now lets explore connection points as ground point types.  A simple scenario of a floating barge 
lifting a sunken barge in 22 feet of water will demonstrate the use of this type of connection. 

If the sunken barge was on a level bottom with its deck 12 feet below the surface, the commands 
to set up the condition and initiate a Multi Body session would be

CLEAR
PROJECT SINKER `this displays in the title bar, helps to keep
READ CSI100.GF `track of which session is which

WEIGHT 300  50  0  6 `weight of 300 Ltons at 50ft aft, on 
`centerline, 6 ft abv the bottom

DR 0 `draft needs to be defined before the ground 
`points are

CRTPT (1) "FWD CONNECTION" 0 20 10 `connection at fwd stbd corner
CRTPT (2) "AFT CONNECTION" 100 20 10 `connection at aft stbd corner

GROUND "FWD STBD" * 5 20 0 /PEN:-22 `at zero draft, 22 ft to the
GROUND "FWD PORT" * 5 -20 0 /PEN:-22 `bottom will put the deck at
GROUND "AFT STBD" * 95 20 0 /PEN:-22 `12 feet below the surface
GROUND "AFT PORT" * 95 -20 0 /PEN:-22

LOAD (*) .95 `to make sure it sinks
SOLVE SECONDARY LIFT.MB  1,1; *,2,2 /WAIT

With the lifting barge directly above the sunken barge at a draft of 2 feet, the commands are

CLEAR
PROJECT FLOATER
READ CSI100.GF

DRAFT 2
SOLVE WE LCG TCG
VCG 6

CRTPT (1) "FWD LIFT POINT" 0 20 10 `lifting at the fwd stbd corner
CRTPT (2) "AFT LIFT POINT" 100 20 10 `lifting at the aft stbd corner

SOLVE PRIMARY LIFT.MB  /WAIT

It is suggested that you type or paste these commands into two run files, for example, 
SINKER.RF for the sunken barge and FLOATER.RF for the floating barge.

It does not matter in this case, which is the primary and which is the secondary as the critical 
point numbers match and both are positive.  But that defines the connections as the latched type. 
If chains or cables are to be used, so the lifting barge can only lift up and not push down, a 
grounding type connection is needed.  To do this, the command for the sunken secondary vessel 
would be

SOLVE SECONDARY LIFT.MB  1,-1; *,2,-2 /WAIT

The second number in each pair corresponds to the secondary vessel.  The negative sign sets 
that point as a grounding point.  As such, it can be pushed up (or in this arrangement, pulled up), 
but it can't be pushed down (or bear weight).

To use load editor and view the Condition Graphic window, add to each run file

VARIABLE CGPARAM="BODY @5 @95,PROFILE"
LOAD (*) EDIT

To initiate a MB session from a single run file, a third run file, such as LIFTBARG.RF, the 
commands

SHELL GHS FLOATER.RF /SPAWN
RUN SINKER.RF

will shell out to another spawned GHS session which will run FLOATER.RF then will run 
SINKER.RF.  Once equilibrium is reached, the Condition Graphic window can be viewed by 
clicking the CG button in Load Editor.

Page 20 of 48



Salvage Procedures using GHS - 4/2011

  

Page 21 of 48

Height of
connection point

Condition Graphics
button

MB flag, green if in
Equilibrium, magenta

If not



Salvage Procedures using GHS - 4/2011

The results will demonstrate two issues.  First, since the connection points were defined at the 
deck of both barges, GHS immediately solved as if the barges were connected at those points, 
pulling the floating barge down while trying to lift the sunken barge. Secondly, if this was the case, 
there would be interference between the two.  To prevent this, addition points can be assign to act 
as constraints.  In this case, adding points to the bottom of the floating barge and connecting 
them to the point on the deck of the sunken barge with the deck of the sunken barge acting as 
ground points. 

To make this a useful MB session, the connection points on the floating barge should be located 
near their respective connection points on the sunken barge.  The deck of the sunken barge we 
know is initially 12 feet below the surface, the deck of the floating barge is 8 feet above the 
surface.  The height of the points on the floating barge need to be decreased 20 feet.  To prevent 
any interference between the hulls, we can add two additional critical points to each.  Two at the 
bottom corners of the floating barge, and two at the top corners of the sunken barge. 
FLOATER.RF should be revised with:

CRTPT (1) "FWD LIFT POINT" 0 20 -10 `10 – 20 = -10
CRTPT (2) "AFT LIFT POINT" 100 20 -10
CRTPT (3) "FWD BOTTOM CORNER" 0 20 0
CRTPT (4) "AFT BOTTOM CORNER" 100 20 0

And to SINKER.RF, add

CRTPT (3) "FWD TOP CORNER" 0 20 10
CRTPT (4) "AFT TOP CORNER" 100 20 10

To prevent interference between the hulls, the MB command for the sunken vessel needs to be 

revised.  If the sunken barge is the secondary vessel, the revision would look like.

SOLVE SECONDARY LIFT.MB  1,-1; *,2,-2; *,-3,3; *,-4,4 /WAIT

The two additional pairs defines the relationship between the new critical points.  The first point in 
each pair corresponds to the primary vessel and the negative sign sets that point on the floating 
vessel as a grounding point.

Rerunning MB should result in little or no reaction forces at the initial connecting points.  The 
sunken barge is grounded on all four points.  The floating barge at at zero or very small heel and 
trim.  To simulate lifting the sunken vessel, the height of the critical points of the floating vessel 
can be increased.

Figure 7

Increase the VCG of the 
Interaction @ 1 to -9 feet.
Enter “+1” or enter “0” then 
“-9”.

As the critical points heights are raised, the floating vessel will increase its trim and heel while the 
sunken vessel may remain on the bottom.  As this happens, the connection points will no longer 
be coincidental.  Recalling from the previous exercise with the barge train, when the connection 
points are not coincidental, discrepancies occur.  The user should be aware of this and attempt to 
keep the differences small.  If this is not possible,  the run should be terminated when the 
differences are large and a new run started with new connection points.

Another issue to be aware of is that using the Load Editor to raise the critical points, only the 
vertical distance relative to the vessel's baseplane was changed.  As it is expected that the 
attaching cables or chain remain perpendicular to the waterplane,  the results are reasonable only 
for small angles of heel and trim.  In the next example, a special form of the CRTPT command is 
used to make adjustments to the connecting point heights perpendicular to the waterplane. 
Unfortunately, this is not possible in Load Editor.

Whether running Multi Body from run files or Load Editor, changes in critical point heights should 
be kept small.  Due to the solving process used by GHS and Multi Body, making a change and 
then changing back to the initial value will not always return to the initial results.  It is important 
that the difference between the heights of the connection point pairs and the magnitude of the 
reaction forces remain small.  Large differences are an indication that the connection points are 
becoming unaligned.

It should be noted that there are times when the using a simple approach may yield better results. 
Taking the above as an example, the realistic approach of two connection points resulted in 
possible errors.  If the sunken barge is actually on a flat bottom, one connection point could be 
used.  Then alignment of the critical points is guaranteed.

An expanded discussion of the workings of MB may be included here in the future.

PREPARING TO RAISE THE SUNKEN FISHING VESSEL
The use of ground points has already been discussed and will not be repeated here.  The number 
and locations will need to be determined by the situation and experience.  The use of the PULL 
command may or may not be appropriate.  This section will discuss the use of Multi Body to roll 
and raise a sunken vessel with two floating barges.

The fishing vessel has sunk and is in 30 feet of water and heeled at 25 degrees to starboard.  Our 
two 100 foot barges are position on each side and will winch the vessel up by cables over the 
side.  Figures 8 and 9 show the arrangement of vessels.

Figure 8

Page 25 of 48



Salvage Procedures using GHS - 4/2011

Figure 9

The following commands will define the initial condition for this example.  The fishing vessel will 
be the secondary vessel as the connections will be grounding type connections and is the 
preferred method.  The weight and center of gravity of the fishing vessel is determined by 
assuming a 10.5 foot initial draft and 1 foot of aft trim.  The VCG is set at 10 feet.  The vessel is 
resting on the bottom with a heel of 25 degrees.  The pilothouse top is set at 5 feet below the 
waterplane.  The ground points used are 3 along the keel and two on the chine.  Or some reason, 
all the hatches and doorways were open so the lazerette, holds, engine room and foc'sle are 
flooded.  These commands should be entered into a run file, FVLIFT.RF

CLEAR
PROJECT FVLIFT
READ FV.GF

TRIM 1/100
DRAFT 10.5
SOLVE WE LCG TCG
VCG 10

HEEL 25
CRTPT (1) "PILOTHOUSE TOP" -24 0 28
HEIGHT (1) -5

GROUND "FWD KEEL" * -27 0 0.157
GROUND "MID KEEL" * -4 0 0
GROUND "AFT KEEL" * 23 0 0
GROUND "FWD CHINE" * -15 10 3
GROUND "AFT CHINE" * 12 12 3

TYPE (LAZ.C, FOCSLE.C, ENGRM.C, HOLD*) FL
SOLVE

STATUS CRT GHS

For lifting points the following critical points will be used.

CRTPT (1) "FWD STBD CONNECTION" -19.25 12 12.90 /NOFLOOD
CRTPT (2) "FWD PORT CONNECTION" -19.25 -12 12.90 /NOFLOOD
CRTPT (3) "AFT STBD CONNECTION" 19.25 12 13.36 /NOFLOOD
CRTPT (4) "AFT PORT CONNECTION" 19.25 -12 13.36 /NOFLOOD

The locations were taken from the geometry file but they do not have to be on the hull.  An 
example of locating them at other locations will be shown below.  Two lifting barges will be used. 
One on each side of the sunken vessel.  The one on the starboard side of the sunken vessel, 
STBDLIFT.RF will be lifting from its port side.  Two starboard tanks will be used for counter 
ballast.  Its initial condition and connection points are as follows:

CLEAR
PROJECT STBDLIFT.RF
READ CSI100.GF
WEIGHT 300 50 0 6
LOAD (TANK2.S,TANK3.S) 0.2

CRTPT (5) "FWD PORT WINCH" 30.75, -20, 11
CRTPT (6) "AFT PORT WINCH" 69.25, -20, 11
SOLVE
ST CRT

The location of the connection points are centered on the barge's length at the same distance 
apart as the points on the sunken vessel, on the barges side and 1 foot above the deck.  If the 
critical point numbers were not specified, they would have been assign 1 and 2 by GHS.  This 
could lead to confusion in the reports, so 5 and 6 was given.

For the other barge positioned on the port side of the fishing vessel, PORTLIFT.RF, a counter 
weight of 25 tons is used located on its port side to give an initial starboard heel.

CLEAR
PROJECT PORTLIFT
READ CSI100.GF
WEIGHT 300 50 0 6
ADD "COUNTER WEIGHT" 25 50 -15 12

CRTPT (7) "FWD STBD WINCH" 30.75, 20, 11
CRTPT (8) "AFT STBD WINCH" 69.25, 20, 11
SOLVE
STATUS CRT

At this step, run each file and note the heights of the critical points.  The points on the barges are 
about 9 feet above the water and the deepest point on the ship is about 24 feet below the surface 
for a total of 33 feet.  The points on the barges will need to be changed to create a connection. 
That could be done by changing the heights in the above statements, but doing so would lower 
them perpendicular to the barge's baseplane as explained earlier.  A better method would be to 
change them using  special form of the critical point command, CRTPT (n) *-d  will lower the 
nth critical point d feet.  The same statement can be used to raise the critical point by replacing 
the minus sign with a plus sign.  The amount each is lowered depends on how the vessel is to be 
lifted.  One option would be to lift it straight upwards with its current heel of 25 degrees, or to 
rolled it upright then lift.  Here we will roll the vessel first.  If it is lifted on the stbd side too far, it 
may tip to port.  To prevent this, we set all the barge's critical points to 33 feet below their present 
position.  Then we will raise each the same amount so the port side connection will support the 
vessel when it is near upright.

To move the critical points, set up the communication channels for Multi body and enter Load 
Editor,  add the following commands

FVLIFT.RF

SHELL GHS STBDLIFT.RF /SPAWN
SHELL GHS PORTLIFT.RF /SPAWN
SOLVE SECONDARY: LIFT1.MB,5,-1; *,6,-3; LIFT2.MB,7,-2; *,8,-4
LOAD (*) EDIT

STBDLIFT.RF

CRTPT (5) *-33
CRTPT (6) *-33
SOLVE PRIMARY:LIFT1.MB
LOAD (*) EDIT

PORTLIFT.RF

CRTPT (3) *-33  
CRTPT (4) *-33
SOLVE PRIMARY:LIFT2.MB
LOAD (*) EDIT

Now executing FVLIFT.RF in GHS will start two additional sessions of GHS/Load Editor.  The 
results can be viewed in the appropriate Weight and Ground Point windows.  The heights of the 
connection points can be changed in Load Editor as in the previous example. But by using run 
files, we can use the CRTPT (n) *-d form of the critical point command.  To do this, the three 
sessions of GHS will need to communicate with each other each time the values are changed.  If 
we step the connection point heights, say one foot at a time, it is necessary that all three sessions 
of GHS reach equilibrium before continuing to the next step.  Since we expect the port side barge 

to not make a connection in the first condition and possibly subsequent conditions, that session 
will need to wait until the others reach equilibrium.  That can be done by looping macros that exit 
when the condition that the other session are ready to continue.

As stated at the beginning of this manual.  It is assumed the user is familiar with macros, user 
variables and passing parameters.  These features of GHS will now be implemented.  For every 
user variable created, they will need to be declared in each run in which they are used.  

Above we defined the two barges each as primary sessions and the sunken vessel as a 
secondary session.  The looping macros will be in the run files associated with the barges.  Then 
neither one will continue until the fishing vessel reaches equilibrium.  The looping macro is written 
as

MACRO IFREADY `creates a looping macro
  WAIT 0.1 `waits for 0.1 second 
  IF {READY}=0 THEN EXIT IFREADY `loop if second session not ready
/

In the run files for each of the primary vessels, the following  macros are needed

MACRO HOIST `to adjust the height of each connection point
  CRTPT (7) *+%1 `note that the critical point numbers must 
  CRTPT (8) *+%2 `correspond to those in the run file
/

MACRO STEP `macro to change point heights
  .HOIST %1 %2 `passes new values to macro hoist
  SOLVE SEND SET NEWLOADP=1 `note, use NEWLOADS=1 for the stbd side
  .IFREADY `starts looping until READY=1 is

`received from secondary session
  CLS
  IF {STARTED}=0 THE SET STARTED=1 ELSE PAGE `for output control
  STATUS CRT GHS
  DI STATUS PROFILE:OUTBOARD, BODY @30.75 @69.25
  SET READY = 0
  WAIT 0.1
/

Variables READY and STARTED need to be declared.  Variables NEWLOADP and NEWLOADS do not 
need to be declared in this run file.  The SOLVE SEND sends what follows to the session with 
which it is communicating.  The command SET NEWLOADP=1 in this case, is only recognized as a 
GHS command in the receiving session.  Therefore, that variable needs to be declared in that run 
file.

To open a report, raise the connection points 8 feet in 1 foot increments and view the results, 

include the following:

REPORT PT_BARGE.PF
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1

WAIT 0.5 `to avoid conflicts with stbd printout
SOLVE SEND SET NEWLOADP=-1 `send to second session to stop macro

`NEWLOADS=-1 for stbd side
REPORT CLOSE /PREVIEW /SPAWN `preview report
END

For the secondary vessel, the following macros are needed

MACRO STEP `macro to start solving for secondary vessel
  WAIT 0.1
  IF {NEWLOADP}<0 THEN IF {NEWLOADS}<0 THEN EXIT
  IF {NEWLOADP}=0 THEN EXIT STEP `if port side not ready, loop
  IF {NEWLOADS}=0 THEN EXIT STEP `if stbd side not ready, loop
  WAIT 0.1 `allow time to detect now unsolved
  SOLVE WAIT:60 `wait until fully solved
  CLS
  IF {STARTED}=0 THEN SET {STARTED}=1 ELSE PAGE
  ST CRT GHS
  DI (*) STATUS BODY @-19.25 @19.25
  SET NEWLOADP=0, NEWLOADS=0
  SOLVE SEND SET READY=1 `broadcasts ready to both port and stbd 
  EXIT STEP `exits macro and reruns macro step
/

REPORT FV_LIFT.PF
.STEP
SOLVE NORMAL `ends Multi Body communications
WAIT 0.5 `avoid conflicts with other printouts
REPORT CLOSE /PREVIEW /SPAWN
END

Variables NEWLOADP, NEWLOADS and STARTED need to be declared.  Due to different 
processor speeds, memory and possibly other issues, the time GHS takes to solve a particular 

condition may vary.  The procedures utilized during Multi Body solving with multiple sessions may 
have timing issues in complicated arrangements.  The WAIT n commands causes timed pauses 
that help to keep the sessions in sequence.  The length of time of the pauses or the need for 
them may vary for different computers or networks if using one.

After successfully running the above run files, three report files will be opened in a preview 
window.  Each report should have 8 pages corresponding to the 8 steps as the connection points 
were changed.  Note that there was no reactions at the connection points on the port barge for 
the first 6 steps.  This is due to the initial condition defined with the critical points on the barge 
positioned below the critical points on the fishing vessel.  All four of the lifting points were raised 1 
foot in each step.  The increments for each step could have different.  For example, the lift points 
for the port barge could have be located at the contact points on the fishing vessel.  This would 
simulate the lines on the port side being snugged up as the starboard barge started to lift to 
prevent an increase of the ground reactions as the vessel rolls.  In this case, initial height of the 
port barge points would be set by 

CRTPT (7) *- 23 `25 degrees of heel puts the port side about
CRTPT (8) *- 23 `10 feet above the stbd side

The steps would then be

.STEP 0 0

.STEP 0 0

.STEP 0 0

.STEP 0 0

.STEP 0 0

.STEP 0 0

.STEP 1 1

.STEP 1 1

Likewise, the steps could be such that only the forward or aft points were raised in a step.

If the operation is carried out as illustrated above, it is apparent that the fishing vessel can not be 
brought to the surface.  In reality, the barges may have sheaves mounted outboard of the hull 
sides, the lifting is done by a crane or the barges are restrained so they are not directly over the 
sunken vessel.  To do this, the connection points can be located outboard of the hull sides.  See 
Figure 10.

Figure 10

By setting up the Multi Body sessions as shown above, the reaction forces given will be 
proportional to the vertical component.  The ratio of the vertical to horizontal components is equal 
to the ratio of vertical to horizontal reactions.  From this, the forces needed to restrain the barges 
and the tension in the chain or cable can be determined.

DEBALLASTING THE SUBMERGE FISHING VESSEL
In some cases, raising a sunken vessel by lifting as above, the vessel can not always be raised to 
a position where enough waterplane area to maintain positive GM while pumping out flooded 
compartments.  In this example, we will assume the fishing vessel has been raised to a level that 
the pilothouse is exposed and the hull is  somewhat accessible.  To empty the flooded 
compartments, we will make use of several of the types of tanks available in GHS.  To begin, the 
initial condition is set using the weight, centers of gravity and lift points as in the example above. 
Using part of the FVLIFT.RF run file, we can quickly start a new run file.  Copy FVLIFT.RF to 
FVDEBALL.RF and make the revisions shown below.  Delete what is crossed out, add what 
highlighted.

CLEAR
PROJECT FVLIFT  FVDEBALL
READ FV.GF

TRIM 1/100
DRAFT 10.5
SOLVE WE LCG TCG
VCG 10

HEEL 25
CRTPT (1) "PILOTHOUSE TOP" -24 0 28
HEIGHT (1) -5  7

GROUND "FWD KEEL" * -27 0 0.157
GROUND "MID KEEL" * -4 0 0
GROUND "AFT KEEL" * 23 0 0
GROUND "FWD CHINE" * -15 10 3
GROUND "AFT CHINE" * 12 12 3

TYPE (LAZ.C, FOCSLE.C, ENGRM.C, HOLD*) FL
SOLVE
STATUS CRT GHS

CRTPT (1) GROUND "FWD STBD CONNECTION" * -19.25 12 12.90 /NOFLOOD
CRTPT (2) GROUND "FWD PORT CONNECTION" * -19.25 -12 12.90 /NOFLOOD
CRTPT (3) GROUND "AFT STBD CONNECTION" * 19.25 12 13.36 /NOFLOOD
CRTPT (4) GROUND "AFT PORT CONNECTION" * 19.25 -12 13.36 /NOFLOOD

MACRO OUTPUT
SO
STATUS WEIGHT, DISPL, WPL:TOTAL
DI (*) STATUS PROFILE
/

REPORT FVDEBALL.PF

.OUTPUT
REPORT /PREVIEW
REPORT OFF

Running the above will produce the follow page

Now that the focsle is exposed, it will be the first to be emptied.  Various methods may be 
employed to do this, for a demonstration of the pressure tank type, we will assume the entry is 
sealed and an opening cut into the aft bulkhead above the main deck.  A connection is made to 
the focsle deck so the compartment can be pressurized.  The pressure tank type using the tank's 
reference point as the opening or breach in the hull.  Add the following to the run file

REPORT FVDEBALL.PF
.OUTPUT

REFPT (FOCSLE.C) -23.1 0 14
CRTPT “FOCLSE OPENING” -23.1 0 14 /NOFLOOD
TYPE (FOCSLE.C) PRESSURE /HEAD:2
.OUTPUT

REPORT /PREVIEW
REPORT OFF

The reference point is set to a meaningful location.  When the PRESSURE tank type is given the 
slash parameter must also be given.  The head given in this case is 2 feet.  GHS will lower the 
level in the compartment to a level 2 feet below the external waterplane.  If the reference point  is 
above the external waterplane, the level will only be lowered to the reference point.  Viewing the 
results shows that the compartment is not complete empty.  Increasing the head to 5 feet will 
lower the level to the reference point.

Now that the engine room vent is almost above the waterplane, cutting an opening in the hull side 
and pressurizing this compartment will empty most of its contents.  Since the location of the 
engine room floor and day tanks are not visible from outside the hull the opening is made above 
and forward of these boundaries.  This and the following additions to the run file should be place 
before the REPORT /PREVIEW command.

REFPT (ENGRM.C) -31.0 6.5 4.5
CRTPT “ENGINE ROOM OPENING” -31.0 6.5 4.5 /NOFLOOD
TYPE (ENGRM.C) PR /HEAD:15
.OUTPUT

The vessel's condition should look like this.

From the above results, it is seen that the opening in the hull side to the engine room is 7.6 feet 
below the surface.  Closing or sealing of the vent opening will trap the air in the compartment.  7.6 
feet of seawater is about 7.6/33 = 0.23 atmospheres.  To make this a bubble tank type with that 
pressure, the type and load commands are used.

TYPE (ENGRM.C) BUBBLE
LOAD (ENGRM.C) /PR:1.23

The above will increase the tank load if the reference point depth increases by balancing the 
pressures at the reference point.  If the reference point depth decreases, the load won't change. 
In reality, air will escape decreasing the pressure, however GHS does not determine this pressure 
drop.  So if the depth increases again, GHS will load the compartment as if the 1.23 atmospheres 
of pressure remains.

At this point, the focsle could be sealed.  To do this while maintaining the amount of residual 
contents trapped in the compartment, add the following

TYPE (FOCSLE.C) DAMAGE
TYPE (FOCSLE.C) INTACT /HBL

By setting the tank type to damage first, the /HBL parameter will keep the trapped contents when 
the tank type is set to intact.  This option is available for damage and flooded tanks.

The next obvious step is to pump out the lazerette.  Using a modified hatch cover with a vent and 
pickup tube, the compartment will be emptied to a point near the hull bottom.  It would be good 
practice to decrease the load in steps to observe the effect of the introduced free surface.  Setting 
the tank type as intact, reduce the load in 25 % increments.

TYPE (LAZ.C) INTACT
LOAD (LAZ.C) 0.75
.OUTPUT
LOAD (LAZ.C) 0.50
.OUTPUT
LOAD (LAZ.C) 0.25
.OUTPUT
LOAD (LAZ.C) 0.05
.OUTPUT

The result is

It is expected (and hoped) that the forward hold will emerge and begin to spill its contents. 
However, the SPILL tank type should not be use here.  A spilling tank maintains the contents 
level at the reference point.  But a spilling tank does so even if the reference point is below the 
external waterplane.  For this situation, the damage tank type is appropriate.  A damaged tank 
with a nominal load of 100% will act as a spilling tank if the reference point is above the external 
waterplane and a flooding tank if the reference point is below the waterplane.  To set the forward 
starboard corner of the hatch opening as the reference point and the tank as damaged, add the 
following 

REFPT (HOLD1.C) -15.5 3 12.5
TYPE (HOLD1.C) DAMAGE
LOAD (HOLD1.C) 1.0

Next, the aft hold is pumped out in increaments.

TYPE (HOLD2.C) INTACT
LOAD (HOLD2.C) 0.75
.OUTPUT
LOAD (HOLD2.C) 0.50
.OUTPUT
LOAD (HOLD2.C) 0.25
.OUTPUT
LOAD (HOLD2.C) 0.05
.OUTPUT

We find that the vessel will be freely floating with the hold pumped out to 75%.  It is left as an 
exercise to determine if the vessel maintains positive stability between full and 75% load.

Appendix – Run files for Examples in this manual

PARBUCK.RF
CLEAR
PROJECT PARBUCK
READ FV.GF

MACRO SHOW
DISP (*) STATUS BODY @ 27F @23.1A @30.8A, PROFILE:OUTBOARD
/

`Define necessary critical points,
`features on the hull that are useful for setting the waterplane.
`Certain ground points and anticipated ground points can also be set.

CRTPT (1) "KEEL AT BOW" 23F 0 0 /NOFLOOD
CRTPT (2) "SKEG AT 23A" 23A 0.6 0 /NOFLOOD
CRTPT (3) "AFT CHINE AT 23A" 23A 11.63 4.78 /NOFLOOD
CRTPT (4) "END OF SKEG" 30.8A 0 0 /NOFLOOD

`Enter a known or estimated lightship weight.
WEIGHT 150 3F 0 11

`Enter added weights, some may have shifted.
ADD "DECK LOAD AGAINST TO BWK" 15 25 10 13
ADD "FISHING GEAR" 5 5 0 20

`Enter any tank loads, weight shifts are calculated by GHS.
LOAD (WT*) .95
LOAD (FODAY*) .95
LOAD (DB1*) .5

`Trim is the perpendicular distance between the waterline
`at two locations. The keel if exposed, parallel deck edges or
`other features may be used.
`If the keel is 0.5 feet higher at 25.0f than at 25.0a,
`then the trim is arctan (0.5/50) = 0.57 degrees aft

TRIM 0.57
`The heel can be estimated from observations.

HEEL 75
`Use the height of a critical point to set the depth.

HEIGHT (4) 2

`Define the ground points already in contact.
`The default penetration is 0, so it does not need to be given.

GROUND "FWD CHINE" * 27F 8.42 4.36
GROUND "FWD HULL SIDE" * 27F 10 10 
GROUND "AFT CHINE" * 23.1A 11.63 4.78
GROUND "AFT HULL SIDE" * 23.1A 11.93 8.89

`For anticipated ground points as the vessel is righted,
`assign ground points on hull with negative penetrations.
`This is the vertical distance from the ground point to the 
`expected point of contact with the bottom as the vessel is righted.

GROUND "KEEL AT BOW" * 23F 0 0 /PEN:-11.5
GROUND "END OF SKEG" * 30.8A 0 0 /PEN:-10

SO
ST GHS
ST CRT
`.SHOW

`Define locations where cables are attached.
PULL "WINCH LINE AT BOW" 0 30.8F 12P 21.2 180 -5
PULL "WINCH LINE AT STERN" 0 23.1A 12.25P 13.55 180 -5

MACRO REHEEL "RESETHEEL"
HEEL 75

Page 39 of 48



Salvage Procedures using GHS - 4/2011
SO
/

`Enter load editor with CG parameter defined. Determine the amount
`of pull force required to right the vessel in the Pull window.
`Macro reheel is to reset the heel if too much pull force is applied.

VARIABLE CGPARAM = "BODY @ 27F @23.1A,BODY @23F @30.8A, PROFILE:OUTBOARD /SYNC"
LOAD (*) EDIT /DISPLAY:OFF /CG /PULL:WINCH LOADS /MACRO:REHEEL, RESULTS

`Knowing about how much pull is needed to "tip" the vessel, about 7 tons
`in this case.

`To determine maximum ground reactions, solve for them over a range of heel.

VARIABLE HE MAXHEEL MAXREACT = 0
.REHEEL
FIX HEEL

REPORT PARBUCK 

MACRO TIPOVER
SET HE = {HEEL} MIN 2
HEEL {HE}
SO
IF {REACT} = 0 THEN EXIT
IF {REACT} > {MAXREACT} THEN SET MAXREACT= {REACT} | SET MAXHEEL = {HEEL}
\TOTAL GROUND REACTION:{ REACT:2},  RA ={ RAH:2},  GMT = { GMT:2}, HEEL ={ HEEL}
/

.TIPOVER (90)

\\\
HEEL {MAXHEEL}
SO
\THE MAXIMUM GROUND REACTIONS OCCUR AT {HEEL} DEGREES\
\

ST DISPL

`to see the righting arm curve
.REHEEL
VARY HEEL
SOLVE
ANGLES 0 -5 -10 ... -90
RAH
REPORT /PREVIEW
REPORT OFF

CAPSIZE.RF
CLEAR
PROJ CAPSIZE
READ SV.GF

VARIABLE CGPARAM = "BODY @-45 @-10, PROFILE /SYNC", HE

`INDENTIFY ATTACH POINTS WITH CRITICAL POINTS
CRTPT OFF
CRTPT (1) "FWD DECK STBD" -45   6.39  9.36 /NOFLOOD
CRTPT (2) "FWD DECK CL  " -45   0     9.60 /NOFLOOD
CRTPT (3) "FWD DECK PORT" -45  -6.39  9.36 /NOFLOOD
CRTPT (4) "FWD BOTT BRIDLE 1"   -45   0      0      /NOFLOOD
CRTPT (5) "FWD BOTT BRIDLE 2"   -45   1      0      /NOFLOOD
CRTPT (6) "FWD BOTT BRIDLE 3"   -45   2.28   0.47   /NOFLOOD
CRTPT (7) "FWD BOTT BRIDLE 4"   -45   3.53   1.86   /NOFLOOD
CRTPT (8) "FWD BOTT BRIDLE 5"   -45   4.25   3.15   /NOFLOOD
CRTPT (9) "FWD BOTT BRIDLE 6"   -45   6.39   9.36   /NOFLOOD

`Set weight, center of gravity and capsize the vessel
WE 2000 -27 0 5
HEEL 180
SO

`define the pull force as a function of heel and location
`the actual force will be passed from the call to the macro

MACRO YANK
PULL "FWD DECK STBD" %1@180 %1@172 0@171 0@0      -45  6.39 9.36 180 10
PULL "FWD DECK CL"   0@180  0@171  %1@170 %1@168  0@167  0@0 -45  0     9.60    180     10
PULL "FWD DECK PORT" 0@180  0@168  %1@167 %1@0      -45 -6.39 9.36 180 10

PULL "FWD AT KEEL 1" %1@180   %1@170    0@169  0@0      -45  -1 0 0 -10
PULL "FWD BOTT 2" 0@180 0@170 %1@169 %1@150 0@149 0@0      -45  1 0 0 -10
PULL "FWD BOTT 3" 0@180 0@150 %1@149 %1@122 0@121 0@0      -45  2.28 0.47 0 -10
PULL "FWD BOTT 4" 0@180 0@122 %1@121 %1@108 0@107 0@0      -45  3.53 1.86 0 -10
PULL "FWD BOTT 5" 0@180 0@108 %1@107 %1@98  0@97 0@0      -45  4.25 3.15 0 -10
PULL "FWD BOTT AT DECK 6" 0@180 0@98 %1@97 %1@0        -45  6.39 9.36 0 -10

`solves and produce a three page report for each step
SO
SET HE = {HEEL}
PULL REPORT
HMMT REPORT
DI STATUS BODY@-45
PAGE
HEEL 0
RA 0 -5 ... -180 /SIZE:1.8
HEEL = {HE}
/

`start a report, run 7 cases with pull forces from 0 to 300 Ltons
REPORT /NOFOOT
.YANK (7,50) 0
REPORT /PREV
REPORT OFF

FWDBARGE.RF
CLEAR
READ CSI100.GF

DR 3
SO WE LCG
VCG 6

CRTPT (1) "STBD PIN" 100 20 10 /NOFLOOD
CRTPT (2) "PORT PIN" 100 -20 10 /NOFLOOD

SOLVE PRIMARY:TRAIN.MB /WAIT

ST GHS
ST CRT

LOAD (*) EDIT /DENFMT:1 /DISPLAY:TRUE

AFTBARGE.RF
CLEAR
READ CSI100.GF

DR 3
SO WE LCG
VCG 6

CRTPT (1) "STBD PIN" 0 18 10 /NOFLOOD
CRTPT (2) "PORT PIN" 0 -18 10 /NOFLOOD

SOLVE SECONDARY:TRAIN.MB,1,1; *,2,2 /WAIT

ST GHS
ST CRT

LOAD (*) EDIT /DENFMT:1 /DISPLAY:TRUE

BARGLIFT.RF
SHELL GHS FLOATER.RF /SPAWN
RUN SINKER.RF

FLOATER.RF
CLEAR
PROJECT FLOATER
READ CSI100.GF

DRAFT 2
SOLVE WE LCG TCG
VCG 6

CRTPT (1) "FWD LIFT POINT" 0 20 -10 `lifting at the fwd stbd corner
CRTPT (2) "AFT LIFT POINT" 100 20 -10 `lifting at the aft stbd corner
CRTPT (3) "FWD BOTTOM CORNER" 0 20 0
CRTPT (4) "AFT BOTTOM CORNER" 100 20 0

SOLVE

SOLVE PRIMARY LIFT.MB  /WAIT

VARIABLE CGPARAM="BODY @5 @95,PROFILE"
LOAD (*) EDIT

SINKER.RF
CLEAR
PROJECT SINKER
READ CSI100.GF

WEIGHT 300  50  0  6 `weight of 300 Ltons at 50ft aft, on 
`centerline, 6 ft abv the bottom

DR 0 `draft needs to be defined before the ground
 `points are
CRTPT (1) "FWD CONNECTION" 0 20 10 `connection at fwd stbd corner
CRTPT (2) "AFT CONNECTION" 100 20 10 `connection at aft stbd corner
CRTPT (3) "FWD TOP CORNER" 0 20 10
CRTPT (4) "AFT TOP CORNER" 100 20 10

GROUND "FWD STBD" * 5 20 0 /PEN:-22 `at zero draft, 22 ft to the 
GROUND "FWD PORT" * 5 -20 0 /PEN:-22 `bottom will put the deck at 
GROUND "AFT STBD" * 95 20 0 /PEN:-22 `12 feet below the surface
GROUND "AFT PORT" * 95 -20 0 /PEN:-22

LOAD (*) .95 `to make sure it sinks
SOLVE

SOLVE SECONDARY LIFT.MB  1,-1; *,2,-2;*,-3,3; *,-4,4 /WAIT

VARIABLE CGPARAM="BODY @5 @95,PROFILE"
LOAD (*) EDIT

FVLIFT.RF
CLEAR
READ FV.GF
VARIABLE NEWLOADP=0, NEWLOADS=0, STARTED=0

`sets up a sunken grounded vessel
TR 1/100
DR 10.5
SO WE LCG TCG
VCG 10

HE 25
CRTPT (1) "PILOTHOUSE TOP" -24 0 28
HEIGHT (1) -5

`crtpt for multi body connections
CRTPT (1) "FWD STBD CONNECTION" -19.25 12 12.90 /NOFLOOD
CRTPT (2) "FWD PORT CONNECTION" -19.25 -12 12.90 /NOFLOOD
CRTPT (3) "AFT STBD CONNECTION" 19.25 12 13.36 /NOFLOOD
CRTPT (4) "AFT PORT CONNECTION" 19.25 -12 13.36 /NOFLOOD

`ground points

GROUND "FWD KEEL" * -27 0 0.157 `/PEN:-15,0.5
GROUND "MID KEEL" * -4 0 0 `/PEN:-15,0.5
GROUND "AFT KEEL" * 23 0 0 `/PEN:-15,0.5
GROUND "FWD CHINE" * -15 10 3 `/PEN:-17,0.5
GROUND "AFT CHINE" * 12 12 3 `/PEN:-17,0.5

TYPE (LAZ.C, FOCSLE.C, ENGRM.C, HOLD*) FL
SO

SHELL GHS STBDLIFT.RF /SPAWN
SHELL GHS PORTLIFT.RF /SPAWN

SOLVE SECONDARY: LIFT1.MB,5,-1; *,6,-3; LIFT2.MB,7,-2; *,8,-4
`sets fv as the secondary vessel
`establishes a channel for communication with other GHS sessions
`1,-5 associates crt 1 from pri with crt 5 from sec, sec vessel grounded
`*,2,-7 same as above, * causes same channel, lift.mb, to be used
`/wait causes execution to pause until a connection is made with at least 
`4 interaction points

MACRO STEP `macro to start solving for secondary vessel
 WAIT 0.1
 IF {NEWLOADP}<0 THEN IF {NEWLOADS}<0 THEN EXIT  `exit if both sides done
 IF {NEWLOADP}=0 THEN EXIT STEP `if portside not ready, loop to top
 IF {NEWLOADS}=0 THEN EXIT STEP `if stbd not ready, loop to top
 WAIT 0.1                       `allow time to detect now unsolved
 SOLVE WAIT:60                  `wait until fully solved
 CLS
 IF {STARTED}=0 THEN SET STARTED=1 ELSE PAGE
 ST CRT GHS
 DI (*) STATUS BODY @-19.25 @19.25
 SET NEWLOADP=0, NEWLOADS=0
 SOLVE SEND SET READY=1         `broadcast ready to both port & stbd
 EXIT STEP `exits macro and reruns macro step
/

REPORT FV_LIFT.PF

.STEP

SOLVE NORMAL

WAIT 0.5                        `avoid conflict with other printouts
REPORT CLOSE /PREVIEW /SPAWN
END

STBDLIFT.RF
CLEAR
READ CSI100.GF
VARIABLE READY=0, STARTED=0
`-----------------------------

WEIGHT 300 50 0 6
LOAD (TANK2.S,TANK3.S) 0.2
`-----------------------------

`initial critical points, should be location of support, sheave, fairlead, etc
CRTPT (5) "FWD PORT WINCH" 30.75, -20, 11
CRTPT (6) "AFT PORT WINCH" 69.25, -20, 11
SOLVE
ST CRT

`adjust the height of the critical points to a height near the connection on the sunken vessel
CRTPT (5) *-33   `<==================  make adjustments here
CRTPT (6) *-33   `<==================  and here

MACRO HOIST
`Crtpts heights to be redefined for each step
CRTPT (5) *+%1
CRTPT (6) *+%2
/

SOLVE PRIMARY:LIFT1.MB `sets barge as the primary vessel
`establishes a channel for communication with other GHS sessions
`/wait causes execution to pause until a connection is made

MACRO IFREADY `creating a looping macro
 WAIT 0.1                       `pauses execution, for 0.1 seconds
 IF {READY}=0 THEN EXIT IFREADY `loop if not ready
/

MACRO STEP `macro to change load of primary vessel
 .HOIST %1 %2                   `sets load in tank %1 to %2
 SOLVE SEND SET NEWLOADS=1      `sends the command set newloads=1 to the second session
 .IFREADY `starts looping until ready=1 is received from second session
 CLS
 IF {STARTED}=0 THEN SET STARTED=1 ELSE PAGE
 STATUS CRT GHS
 DI STATUS PROFILE:OUTBOARD, BODY @30.75 @69.25
 SET READY = 0
 WAIT 0.1
/

REPORT ST_BARGE.PF

`Hoisting sequence
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1

SOLVE SEND SET NEWLOADS=-1

REPORT CLOSE /PREVIEW /SPAWN
END

PORTLIFT.RF
CLEAR
READ CSI100.GF
VARIABLE READY=0, STARTED=0
`-----------------------------

WEIGHT 300 50 0 6
ADD "COUNTER WEIGHT" 25 50 -15 12
`-----------------------------

`initial critical points required to establish communication
CRTPT (7) "FWD STBD WINCH" 30.75, 20, 11
CRTPT (8) "AFT STBD WINCH" 69.25, 20, 11
SOLVE
ST CRT

`adjust the height of the critical points to a height near the connection on the sunken vessel
CRTPT (7) *-33   `<==================  make adjustments here
CRTPT (8) *-33   `<==================  and here

MACRO HOIST
`Crtpts heights to be redefined for each step
CRTPT (7) *+%1
CRTPT (8) *+%2
/

SOLVE PRIMARY:LIFT2.MB `sets barge as the primary vessel
`establishes a channel for communication with other GHS sessions
`/wait causes execution to pause until a connection

MACRO IFREADY `creating a looping macro
 WAIT 0.1                       `pauses execution, for 0.1 seconds
 IF {READY}=0 THEN EXIT IFREADY `loop if not ready
/

MACRO STEP `macro to change load of primary vessel
 .HOIST %1 %2                   `sets load in tank %1 to %2
 SOLVE SEND SET NEWLOADP=1      `sends the command set newloadp=1 to the second session
 .IFREADY `starts looping until ready=1 is received from second session
 CLS
 IF {STARTED}=0 THEN SET STARTED=1 ELSE PAGE
 STATUS CRT GHS
 DI STATUS PROFILE:OUTBOARD, BODY @30.75 @69.25
 SET READY = 0
 WAIT 0.1
/

REPORT PT_BARGE.PF

`Hoisting sequence
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1
.STEP 1 1

WAIT 0.5                        `avoid conflict with stbd side printout
SOLVE SEND SET NEWLOADP=-1

REPORT CLOSE /PREVIEW /SPAWN
END

FVDEBALL.RF
CLEAR
PROJECT FVDEBALL
READ FV.GF

`sets up the fishing vessel supported by the ground points
`used in the MB lifting example
TR 1/100
DR 10.5
SO WE LCG TCG
VCG 10

TYPE (LAZ.C, FOCSLE.C, ENGRM.C, HOLD*) FL

CRTPT (5) "PILOTHOUSE TOP" -24 0 28
HEIGHT (5) 7

`crtpt from multi body connections
GROUND "FWD STBD CONNECTION" * -19.25 12 12.90
GROUND "FWD PORT CONNECTION" * -19.25 -12 12.90
GROUND "AFT STBD CONNECTION" * 19.25 12 13.36
GROUND "AFT PORT CONNECTION" * 19.25 -12 13.36

MACRO OUTPUT
SO
STATUS CRT GHS
DI (*) STATUS PROFILE
/

REPORT FVDEBALL.PF
.OUTPUT

REFPT (FOCSLE.C) -23.1 0 14
CRTPT "FOCLSE OPENING" -23.1 0 14 /NOFLOOD
TYPE (FOCSLE.C) PR /HEAD:5
.OUTPUT

REFPT (ENGRM.C) -30 6.5 4.5
CRTPT "ENGINE ROOM OPENING" -30 6.5 4.5 /NOFLOOD
TYPE (ENGRM.C) PR /HEAD:15
.OUTPUT

TYPE (ENGRM.C) BUBBLE
LOAD (ENGRM.C) * /PR:1.23

TYPE (FOCSLE.C) DAMAGE
TYPE (FOCSLE.C) INTACT /HBL

REFPT (HOLD1.C) -15.5 3.5 12.5
TYPE (HOLD1.C) DAMAGE
LOAD (HOLD1.C) 1.0

TYPE (LAZ.C) INTACT
LOAD (LAZ.C) 0.75
.OUTPUT
TYPE (LAZ.C) INTACT
LOAD (LAZ.C) 0.50
.OUTPUT
TYPE (LAZ.C) INTACT
LOAD (LAZ.C) 0.25
.OUTPUT
TYPE (LAZ.C) INTACT
LOAD (LAZ.C) 0.05
.OUTPUT

TYPE (HOLD2.C) INTACT
LOAD (HOLD2.C) 0.75
.OUTPUT
LOAD (HOLD2.C) 0.50

.OUTPUT
LOAD (HOLD2.C) 0.25
.OUTPUT
LOAD (HOLD2.C) 0.05
.OUTPUT

REPORT /PREVIEW
REPORT OFF

